
1 Introduction

When I was a student, Simula was one of the languages taught in introductory pro-
gramming language courses and I vividly remember a sticker one of our instructors
had attached to the door of his office, saying “Simula does it with class”. I guess
the same holds for Haskell except that Haskell replaces classes by type classes.

Armed with singleton types, multiple-parameter type classes, and functional de-
pendencies we reconsider a problem raised and solved by Danvy in a previous pearl
(1998). The challenge is to implement a variant of C’s printf function, called format
below, in a statically typed language. Here is an interactive session that illustrates
the problem.

Main〉 :type format (lit "helloÃworld")
String
Main〉 format (lit "helloÃworld")
"helloÃworld"

Main〉 :type format int
Int → String
Main〉 format int 5
"5"

Main〉 :type format (int ^ lit "ÃisÃ" ^ str)
Int → String → String
Main〉 format (int ^ lit "ÃisÃ" ^ str) 5 "five"

"5ÃisÃfive"

The format directive lit s means emit s literally. The directives int and str instruct
format to take an additional argument of the types Int and String respectively,
which is then shown. The circumflex ‘^’ is used to concatenate two directives.

The type of format depends on its first argument, the format directive. In a
language with dependent types, such as Cayenne (Augustsson, 1999), format is
straightforward to implement. This pearl shows that format is equally straightfor-
ward to realize in a language, such as Haskell, that allows the definition of values
that depend on types. Our solution enjoys nice algebraic properties and is more
direct than the original one (the relation between the two approaches is detailed in
Sec. 5).

2 Preliminaries: functors

This section briefly reviews the categorical concept of a functor, which is at the
heart of the Haskell implementation of format . For our purposes it is sufficient to
think of a functor as a combination of a type constructor F of kind ? → ? and a so-
called mapping function that lifts a given function of type A → B to a function of
type F A → F B . In Haskell, the concept of a functor is captured by the following
class definition.1

class Functor F where
map :: (A → B) → (F A → F B)

Instances of this class are supposed to satisfy the two functor laws:

map id = id

map (φ · ψ) = map φ · map ψ.

Typical examples of functors are container types such as lists or trees. In these
cases, the mapping function simply applies its first argument to each element of a
given container leaving its structure intact. However, the notion of functor is by no
means restricted to container types. For instance, the functional type (A →) for
fixed A is a functor with the mapping function given by post-composition.2

instance Functor (A →) where
map φ x = φ · x

For this instance, the functor laws reduce to id · x = x and (φ · ψ) · x = φ · (ψ · x ).
The functor (A →) will play a prominent rôle in the following sections. In addition,
we require the identity functor and functor composition.

type Id A = A

instance Functor Id where
map = id

type (F · G) A = F (G A)

instance (Functor F ,Functor G) ⇒ Functor (F · G) where
map = map · map

Again, it is easy to see that the functor laws are satisfied. Furthermore, functor
composition is associative and has the identity functor as a unit. As an aside,
note that these instance declarations are not legal Haskell since Id and ‘·’ are not
data types defined by data or by newtype. A data type, however, introduces
an additional data constructor which affects the readability of the code. Instead
we employ type declarations as if they worked as newtype declarations. Sec. 6
describes the necessary amendments to make the code run under GHC or Hugs.

1 We slightly deviate from Haskell’s lexical syntax: both type constructors and type variables are
written with an initial upper-case letter (a type variable typically consists of a single upper-case
letter) and both value constructors and value variables are written with an initial lower-case
letter. This convention helps us to keep values and types apart.

2 The so-called operator section (A →) denotes the partial application of the infix operator ‘→’
to A.

3 Functional unparsing

The Haskell solution is developed in two steps. In this section we show how to
define format as a type-indexed value. The following section then explains how
to implement the type-indexed value using multiple-parameter type classes with
functional dependencies.

Recall that the type of format depends on its first argument, the format direc-
tive. Clearly, we cannot define such a dependently typed function in Haskell if we
represent directives by elements of a single data type, say,

data Dir = lit String | int | str | Dir ^ Dir .

However, using Haskell’s type classes we can define values that depend on types. In
order to utilize this feature we must arrange that each directive possesses a distinct
type. To this end we introduce the following singleton types:

data LIT = lit String
data INT = int
data STR = str
data D1 ^ D2 = D1 ^ D2.

Strictly speaking, LIT is not a singleton type since it accommodates more than one
element. This is unproblematic, however, since the type of format does not depend
on the argument of lit . Given these declarations, the directive int ^ lit "ÃisÃ"^ str ,
for instance, has type INT ^ LIT ^ STR: the structure of the directive is mirrored
on the type level. As an aside, note that the type constructor ‘^’, which takes
singleton types to singleton types, is isomorphic to the type of pairs. We could have
used pairs in the first place but the right-associative infix data constructor ‘^’ saves
some parentheses.

We can now define format as a type-indexed value of type

formatD :: D → FormatD String ,

that is, formatD takes a directive of type D and returns ‘something’ of String where
‘something’ is determined by D in the following way.

FormatD :: ? → ?

FormatLIT S = S
FormatINT S = Int → S
FormatSTR S = String → S
FormatD1^D2 S = FormatD1 (FormatD2 S )

The type FormatD is a so-called type-indexed type, a type that depends on a type.
It specifies for each of the directives the additional argument(s) format has to take.
The most interesting clause is probably the last one: the arguments to be added
for D1 ^ D2 are the arguments to be added for D1 followed by the arguments to
be added for D2. The crucial property of FormatD is that it constitutes a functor.

This can be seen more clearly if we rewrite FormatD in a point-free style.

FormatLIT = Id
FormatINT = (Int →)
FormatSTR = (String →)
FormatD1^D2 = FormatD1 · FormatD2

The implementation of format is straightforward except perhaps for the last case.

formatD :: D → FormatD String
formatLIT (lit s) = s
formatINT int = λi → show i
formatSTR str = λs → s
formatD1^D2

(d1 ^ d2) = formatD1
d1 ¦ formatD2

d2

So formatINT int is just the show function and formatSTR str is just the identity
on String . It remains to define the operator ‘¦’, which takes an F String and a
G String to a (F · G) String . We know that F = FormatD1 and G = FormatD2

but this does not get us any further. The only assumption we may safely take is
that F and G are functorial. Fortunately, using the mapping function on F we can
turn a value of type F String into a value of type F (G String) provided we supply
a function that takes a String , say, s to a value of type G String . We can define a
function of the desired type using the mapping function on G provided we supply
a function that takes a string, say, t to some resulting string. Now, since we have to
concatenate the ‘output’ produced by the two arguments of ‘¦’, the resulting string
must be s ++ t .

(¦) :: (Functor F ,Functor G) ⇒ F String → G String → (F · G) String
f ¦ g = map (λs → map (λt → s ++ t) g) f

The operator ‘¦’ enjoys nice algebraic properties: it is associative and has the empty
string, "" :: Id String , as a unit. The proof of these properties makes use of the
functor laws and the fact that (String , ++, "") forms a monoid. That said it becomes
clear that the construction can be readily generalized to arbitrary monoids. As an
example, for reasons of efficiency one might want to replace (String , ++, "") by
(ShowS , ·, id), which features constant-time concatenation.

4 Functional unparsing in Haskell

How can we implement the type-indexed value formatD ::D → FormatD String us-
ing Haskell’s type classes? Clearly, a singleton parameter class won’t do since both D
and FormatD vary. We are forced to introduce a two argument class that addition-
ally abstracts away from FormatD assigning format the general type D → F String .
This type is, however, too general since now D and F may vary independently of
each other. This additional ‘flexibility’ is, in fact, not very welcome since it gives
rise to severe problems of ambiguity. Fortunately, functional dependencies (Jones,

2000) save the day as they allow us to capture the fact that F is determined by D .

class (Functor F ) ⇒ Format D F | D → F where
format :: D → F String

The functional dependency D → F (beware, this is not the function space arrow)
constrains the relation to be functional: if both Format D1 F1 and Format D2 F2

hold, then D1 = D2 implies F1 = F2. Note that F is additionally restricted to
be an instance of Functor . It remains to supply for each directive D an instance
declaration of the schematic form instance Format D (FormatD) where format =
formatD .

instance Format LIT Id where
format (lit s) = s

instance Format INT (Int →) where
format int = λi → show i

instance Format STR (String →) where
format str = λs → s

instance (Format D1 F1,Format D2 F2) ⇒ Format (D1 ^ D2) (F1 · F2) where
format (d1 ^ d2) = format d1 ¦ format d2

In implementing the specification of Sec. 3 we have simply replaced a type function
by a functional type relation. Before we proceed let us take a look at an example
translation.

format (int ^ lit "ÃisÃ" ^ str)

= { definition of format }
show ¦ "ÃisÃ" ¦ id

= { definition of ‘¦’ }
mapInt→ (λs → mapId (λt → mapString→ (λu → s ++ t ++ u) id) "ÃisÃ") show

= { definition of mapA→ and mapId }
(λs → (λt → (λu → s ++ t ++ u) · id) "ÃisÃ") · show

= { algebraic simplifications and β-conversion }
λi → λu → show i ++ "ÃisÃ" ++ u

We obtain exactly the function one would have written by hand. Note that simpli-
fications along these lines can always be performed at compile time since the first
argument of format is essentially static (apart from lit ’s string argument).

5 Back to continuation-passing style

It is instructive to compare our solution to the original one by Danvy (1998), which
makes use of a continuation and an accumulating argument. Phrased as a Haskell
type class Danvy’s solution reads:

class Format ′ D F | D → F where
format ′ :: ∀A .D → (String → A) → (String → F A)

instance Format ′ LIT Id where
format ′ (lit s) = λκ out → κ (out ++ s)

instance Format ′ INT (Int →) where
format ′ int = λκ out → λi → κ (out ++ show i)

instance Format ′ STR (String →) where
format ′ str = λκ out → λs → κ (out ++ s)

instance (Format ′ D1 F1,Format ′ D2 F2) ⇒ Format ′ (D1 ^ D2) (F1 · F2) where
format ′ (d1 ^ d2) = λκ out → format ′ d1 (format ′ d2 κ) out

format :: (Format ′ D F ) ⇒ D → F String
format d = format ′ d id ""

Two remarks are in order. First, the instances do not require mapping functions,
which explains why Format ′ is not declared a subclass of Functor , though morally
the second argument of Format ′ is a functor. Second, format ′ (d1 ^ d2) can be
simplified to format ′ d1 · format ′ d2, where ‘·’ is ordinary function composition. We
will take these points up again in the following section.

So we interpret format d by values of type F String whereas Danvy employs
values of type ∀A . (A → String) → (A → F String). An obvious question is, of
course, whether the two approaches are equivalent. Here are functions that convert
to and fro:

α d = λκ out → map (λs → κ (out ++ s)) d
γ d ′ = d ′ id "".

The coercion function α introduces a continuation and an accumulating string,
while γ supplies an initial continuation and an empty accumulating string.

It is easy to see that γ · α = id :

γ (α d) = { definition of γ and α }
(λκ out → map (λs → κ (out ++ s)) d) id ""

= { β-conversion }
map (λs → id ("" ++ s)) d

= { algebraic simplifications }
map id d

= { functor laws }
d

When we try to prove the converse, α · γ = id ,

α (γ d ′) = { definition of α and γ }
λκ out → map (λs → κ (out ++ s)) (d ′ id ""),

we are immediately stuck. There is no obvious way to simplify the final expression.
Note, however, that d ′ has a polymorphic type, so we can appeal to the parametric-

ity theorem. The ‘free theorem’ for d ′ :: ∀A . (String → A) → (String → F A) is
that for all φ :: A1 → A2 and for all ε :: String → A1,

map φ · d ′ ε = d ′ (φ · ε). (1)

Loosely speaking, this rule allows us to shift a part of the continuation to the left.
Continuing the proof we obtain:

= { parametricity (1): φ = λs → κ (out ++ s) and ε = id }
λκ out → d ′ (λs → κ (out ++ s)) "".

We are stuck again. This time we require a rule that allows us to shift a part of the
continuation to the right. Let us assume for the moment that for all ε ::String → A
and for all σ :: String → String ,

d ′ (ε · σ) = d ′ ε · σ. (2)

Given this property, we can finish the proof:

= { proof obligation (2): ε = κ and σ = λs → out ++ s }
λκ out → d ′ (λs → κ s) (out ++ "")

= { algebraic simplifications and η-conversion }
d ′

It remains to establish the proof obligation. Perhaps unsurprisingly, it turns out
that the rule does not hold in general. The problem is that the accumulating argu-
ment has a too concrete type: it is a string, which we can manipulate at will. In the
following instance, for example, the accumulator is replaced by an empty string.

data CANCEL = cancel

instance Format ′ CANCEL Id where
format ′ cancel = λκ out → κ ""

The effect of cancel is to discard the string produced by the directives to its left.

Main〉 format ′ (int ^ lit "ÃisÃ" ^ cancel ^ str) 5 "five"

"five"

One might argue that the ability to define such a directive is an unwanted conse-
quence of switching to continuation passing style. In that sense, rule (2) is really a
proof obligation for the programmer. As a closing remark, note that we can achieve
a similar effect in our setting using a ‘forgetful’ variant of ‘¦’:

f . g = map (λs → map (λt → t) g) f
f / g = map (λs → map (λt → s) g) f .

6 Applying a functor

Let us finally turn the code of Sec. 4 into an executable Haskell program. Recall
that the instance declarations involving the type synonyms Id and ‘·’ are not legal
since type synonyms must not be partially applied. Therefore, we are forced to
introduce the two types via newtype declarations:

newtype Id A = ide A

newtype (F · G) A = com (F (G A)).

Alas, now Id and ‘·’ are new distinct types. In particular, the identities Id A = A and
(F · G) A = F (G A) do not hold any more: the type of format (int^lit "ÃisÃ"^str)
is ((Int →) · Id · (String →)) String rather than Int → String → String . In order
to obtain the desired type we have to apply the functor (Int →) · Id · (String →)
to the type String . This type transformation is implemented by the following three
parameter type class.

class (Functor F ) ⇒ Apply F A B | F A → B where
apply :: F A → B

instance Apply (A →) B (A → B) where
apply = id

instance Apply Id A A where
apply (ide a) = a

instance (Apply G A B ,Apply F B C ) ⇒ Apply (F · G) A C where
apply (com x ) = apply (map apply x )

The intention is that the type relation Apply F A B holds iff F A = B . Conse-
quently, B is uniquely determined by F and A, which is expressed by the func-
tional dependency F A → B (again, do not confuse the dependency with a func-
tional type). The class method apply always equals the identity function since a
newtype has the same representation as the underlying type. Now, renaming the
class method of Format to formatx we arrive at the true definition of format :

format :: (Format D F ,Apply F String A) ⇒ D → A
format d = apply (formatx d).

7 Haskell can do it (almost) without type classes

Given the title of the pearl this final twist is perhaps unexpected. We can quite
easily eliminate the Format class by specializing format to the various types of
directives: for each d ::D we introduce a new directive d ::FormatD String given by
d = formatx d—we omit the underlining in the sequel and just reuse the original
names.

lit :: String → Id String
lit s = ide s

int :: (Int →) String
int = λi → show i

str :: (String →) String
str = λs → s

format :: (Apply F String A) ⇒ F String → A
format d = apply d

formatIO :: (Apply F (IO ()) A) ⇒ F String → A
formatIO d = apply (map putStrLn d)

So int is just show (albeit with a less general type), str is just id , and format is
just apply (again with a less general type). Furthermore, instead of ‘^’ we use ‘¦’.

We have also defined a variant of format that outputs the string to the standard
output device. This function nicely demonstrates how to define one’s own variable-
argument functions on top of format . Here is an example session that illustrates
the use of the new unparsing combinators.

Main〉 :type (int ¦ lit "ÃisÃ" ¦ str)
((Int →) · Id · (String →)) String
Main〉 :type format (int ¦ lit "ÃisÃ" ¦ str)
Int → String → String
Main〉 format (int ¦ lit "ÃisÃ" ¦ str) 5 "five"

"5ÃisÃfive"

Main〉 format (show ¦ lit "ÃisÃ" ¦ show) 5 "five"

"5ÃisÃ\"five\""

Main〉 format (lit "sumÃ" ¦ show ¦ lit "Ã=Ã" ¦ show) [1 . . 10] (sum [1 . . 10])
"sumÃ[1,2,3,4,5,6,7,8,9,10]Ã=Ã55"

Note the use of show in the last two examples—but, please, don’t ask for the type of
format (show ¦ lit "ÃisÃ"¦show). In fact, we can now seamlessly integrate Haskell’s
predefined unparsing function with our own routines. As an illustration, consider
the following directive for unparsing a list of values.

list :: (A →) String → ([A] →) String
list d [ ] = "[]"

list d (a:as) = "[" ++ d a ++ rest as
where rest [ ] = "]"

rest (a:as) = ",Ã" ++ d a ++ rest as

To format a string, for instance, we can now either use the directive str (emit the
string literally), show (put the string in quotes), or list show (show the string as a
list of characters). Likewise, for formatting a list of strings we can choose between
show , list str , list show , or list (list show).

Can we also get rid of Id , ‘·’ and consequently of the class Apply? Unfortunately,
the answer is in the negative. Though all directives possess legal Haskell 98 types,
Haskell’s kinded first-order unification gets in the way when we combine the di-
rectives. Loosely speaking, the newtype constructors are required to direct the
type checker. Interestingly, Danvy’s solution seems to require a less sophisticated
type system: the combinators possess ordinary Hindley-Milner types. However, this
comes at the expense of type safety as a closer inspection reveals. The critical com-
binator is the one for concatenating directives, which possesses the following rank-2

type (consider the instance declaration for ‘^’ in Sec. 5).

(·) :: ∀F G . (∀X . (String → X ) → (String → F X ))
→ (∀Y . (String → Y ) → (String → G Y ))
→ (∀Z . (String → Z ) → (String → F (G Z )))

Since ‘·’ amounts to function composition we can generalize (or rather, weaken) the

type to
(·) :: ∀A B C . (B → C )

→ (A → B)
→ (A → C ).

Since Danvy’s combinators furthermore do not employ mapping functions, they
can be made to run in a language with a Hindley-Milner type system. Or course,
weakening the types has the immediate drawback that, for instance, the non-sensible
call format (const · length · run) where run k = k "" is well-typed.

Stephanie Weirich for their comments on a previous draft of this paper.

References

Augustsson, L. (1999) Cayenne – a language with dependent types. SIGPLAN Notices
34(1):239–250.

Danvy, O. (1998) Functional unparsing. J. Functional Programming 8(6):621–625.

Jones, M. P. (2000) Type classes with functional dependencies. Smolka, G. (ed), Proceed-
ings of the 9th European Symposium on Programming, ESOP 2000, Berlin, Germany.
Lecture Notes in Computer Science 1782, pp. 230–244. Springer-Verlag.


